Equação Diferencial Ordinária Linear

Solução analítica de equações diferenciais ordinárias, lineares, a parâmetros constantes e de 1ª ordem:
Apenas para ilustração, vamos apresentar formas de resolver analiticamente a equação diferencial de 1ª ordem:

$$\dot{x} - ax = bu$$

A entrada u é um degrau unitário (ou seja $u = 1$ para $t > 0$) e a condição inicial é $x(0) = 0$.
A solução é uma combinação linear da solução da equação homogênea com uma solução particular.

Equação homogênea:

$$\dot{x} - ax = 0$$

A solução da equação homogênea é do tipo $x_h(t) = e^{\lambda t}$, onde λ é a raiz da equação característica:

Equação característica: $\lambda - a = 0 \quad \Rightarrow \quad \lambda = a$

De fato, substituindo $x_h(t) = e^{\lambda t} = e^{at}$ na equação diferencial:

$$\dot{x}_h - ax_h = 0 \qquad \text{verificando: } \dot{x}_h - ax_h = \frac{d}{dt}\left(e^{at}\right) - a.e^{at} = a.e^{at} - a.e^{at} = 0$$

Portanto a solução proposta satisfaz a equação homogênea.

Solução particular, lembrando que a entrada u é um degrau unitário (ou seja $u = 1$ para $t > 0$):

$$\dot{x} - ax = bu \underset{\text{como } u=1}{\Longrightarrow} \dot{x} - ax = b$$

Uma solução particular é $x_p(t) = -\dfrac{b}{a}$. Substituindo na equação diferencial:

$$\dot{x}_p - ax_p = b \qquad \text{verificando: } \dot{x}_p - ax_p = \frac{d}{dt}\left(-\frac{b}{a}\right) - a.\left(-\frac{b}{a}\right) = 0 + b = b$$

Portanto a solução particular proposta satisfaz a equação diferencial.

A solução da equação é uma combinação linear da solução da equação homogênea e da solução particular:

$$x(t) = C_1 x_h(t) + C_2 x_p(t)$$

$$x(t) = C_1 e^{at} + C_2\left(-\frac{b}{a}\right)$$

Substituindo na equação diferencial:

$$\dot{x} - ax = b \Rightarrow \frac{d}{dt}\left[C_1 e^{at} - C_2\frac{b}{a}\right] - a\left[C_1 e^{at} - C_2\frac{b}{a}\right] = b \Rightarrow aC_1 e^{at} - aC_1 e^{at} + aC_2\frac{b}{a} = b$$

Para satisfazer a equação temos que:

$$aC_2\frac{b}{a} = b \quad \Rightarrow \quad C_2 = 1$$

O valor de C_1 pode ser qualquer, entretanto, para satisfazer a condição inicial:

$$x(0) = 0 \Rightarrow x(0) = C_1 e^{at} + C_2\left(-\frac{b}{a}\right) = C_1 e^{a0} + 1\left(-\frac{b}{a}\right) = C_1 - \frac{b}{a} = 0$$

$$C_1 = \frac{b}{a}$$

Portanto a solução da equação é:

$$x(t) = \frac{b}{a}e^{at} - \frac{b}{a} \quad \Rightarrow \quad x(t) = -\frac{b}{a}\left(1 - e^{at}\right)$$

Observe que, no exemplo do reservatório, a é um número negativo, portanto a parcela e^{at} vai diminuindo exponencialmente com o tempo. Se o valor de a fosse positivo, a parcela e^{at} aumentaria indefinidamente, caracterizando um sistema instável. Pode-se demonstrar que, para uma equação diferencial ordinária linear a parâmetros constantes, se as raízes da equação característica tiverem parte real negativa, a solução da equação diferencial é limitada para uma entrada limitada, ou seja, o sistema é estável.

Simulação numérica de sistema linear

Um sistema linear pode ser representado por:

$\dot{\mathbf{x}} = \mathbf{Ax} + \mathbf{Bu}$ (equações diferenciais – equações de estado)

$\mathbf{y} = \mathbf{Cx} + \mathbf{Du}$ (equações algébricas – equações da saída)

No caso do exemplo do reservatório:

$$\dot{x} = \underbrace{-\frac{1}{2S}\sqrt{\frac{\rho g}{R h_o}}}_{A}x + \underbrace{\frac{1}{S}}_{B}u \qquad \text{(equações diferenciais)}$$

$$y = \underbrace{+1}_{C}x + \underbrace{0}_{D}u \qquad \text{(equações algébricas)}$$

A solução numérica da equação diferencial ordinária, linear, a parâmetros constantes pode ser determinada no Scilab da seguinte forma:

Em um programa de edição de texto sem formatação, como o Notepad, deve se escrever um texto, com o nome "**simulinear.sce**", com os comando que implementam a simulação, e salvando na pasta (por exemplo, no Windows) "**C:\Documents and Settings\Administrador\Desktop\Curso**":

```
// Simulacao de sistema linear
// Eh sempre melhor apagar as variaveis anteriores
clear all

// Definir parametros:
S=10;          // [m^2] Area da secao transversal do reservatorio
rho=1000;      // [kg/m^3] massa especifica da agua
g=10;          // [m/s^2] aceleração da gravidade na superficie da Terra
R=2*10^8;      // [Pa/(m^3/s)^2] parametro que relaciona pressao e vazao
ho=2;          // [m] nivel do reservatorio em regime
hi=0.1;        // [m] nivel adicional desejado
Qei=(1/2)*sqrt(rho*g/(R*ho))*hi;  // [m^3/s] vazao na entrada

// Definir o sistema linear usando o comando syslin:
A=(-1/(2*S))*sqrt(rho*g/(R*ho));
B=1/S;
C=1;
D=0;
tanque=syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema eh
                            // continuo no tempo

// Definir a condicao inicial:
x0=0;          // [m] desvio inicial do nivel em relação ao equilibrio

// Definir o vetor de instantes de tempo:
t=0:10:40000;

// Definir o vetor de entradas:
u=Qei*ones(t);

// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque,x0);

// Plotando o resultado em verde:
plot2d(t,y,3)

// Colocando um titulo na figura e nomeando os eixos:
xtitle("Resposta do tanque","Tempo t [s]","Variacao de nivel [m]");

// Colocando uma grade azul no grafico:
xgrid(2)
```

Para realizar a simulação, abre-se o Scilab, e se executa o seguinte comando:

```
exec('C:\Documents and Settings\Administrador\Desktop\Curso\simulinear.sce')
```

Exercícios:

1 - Faça as modificações adequadas para se poder desenhar e comparar os gráficos da resposta do sistema não linear e linear. Faça as simulações dos sistemas linear e não linear considerando que o reservatório parte do nível h = 2 m, mas com vazão de entrada nula. Compare as respostas.

2 - Obtenha o modelo matemático do circuito elétrico mostrado abaixo e compare com o modelo linear do sistema com um reservatório. Faça simulações e compare qualitativamente com os resultados do exercício 1 (sistema linear).



<u>Lição de casa</u>:

1 - Usando a abordagem vista nestes exemplos, faça a simulação do sistema com dois reservatórios, supondo o modelo linear:



2 - Desenvolva um circuito elétrico análogo ao sistema com dois reservatórios.

3 - Estude a Lista E.