Modelagem e Análise de um Sistema de Suspensão

Estude o seguinte exemplo:

m = 1 kg
b = 10 N.s/m
k = 900 N/m

A entrada é a velocidade $v_G(t)$ e a saída é a deflexão $x(t)$ da mola.



Este modelo em particular é a parâmetros concentrados, e o modelo será um sistema de equações diferenciais ordinárias lineares a parâmetros constantes.

Equações de estado:
Variáveis de estados:
$v(t)$: velocidade vertical do bloco (variável relacionada com a energia cinética do bloco).
$x(t)$: deflexão da mola (variável relacionada com a eneria potencial elástica da mola).

Entradas:
$v_G(t)$: velocidade vertical do ponto de veículo em contato com solo.

Saída:
$x(t)$: deflexão da mola.

Parâmetros:
m : massa do bloco.
k : rigidez da mola (constante elástica).
b : constante do amortecimento.

Equações de estado:
Pela cinemática sabemos que:

$$\dot{x} = v - v_G$$

Usando o Teorema do Movimento do Baricentro:

$$m\dot{v} = -F_{mola} - F_{amortecedor} - mg$$

Sabemos ainda que:

$$F_{mola} = kx$$

E que:

$$F_{amortecedor} = b(v - v_G)$$

Portanto:

$$m\dot{v} = -kx - b(v - v_G) - mg$$

Ponto de equilíbrio:
Vamos considerar $v_G = 0$ no equilíbrio. Por definição, no equilíbrio as derivadas no tempo das variáveis de estado são nulas:

$$\dot{x} = 0$$

$$v_e - v_{Ge} = 0$$

$$v_e = 0$$

$$m\dot{v} = 0$$

$$-kx_e - b(v_e - v_{Ge}) - mg = 0$$

$$kx_e = -mg \quad \Rightarrow \quad x_e = -\frac{mg}{k}$$

Desvios em relação ao equilíbrio:

$$x_1 = x - x_e \quad \Rightarrow \quad x = x_1 + x_e$$

$$x_2 = v - v_e \quad \Rightarrow \quad v = x_2 + v_e$$

$$u = v_G - v_{Ge} \quad \Rightarrow \quad v_G = u + v_{Ge}$$

Podemos reescrever as equações da seguinte forma:

$$\dot{x} = v - v_G \quad \Rightarrow \quad \frac{d}{dt}(x_1 + x_e) = (x_2 + v_e) - (u + v_{Ge})$$

$$\dot{x}_1 + \dot{x}_e = x_2 - u + v_e - v_{Ge}$$

Como x_e é constante, $\dot{x}_e = 0$, e verificamos também que $v_e = 0$ e $v_{Ge} = 0$, logo:

$$\dot{x}_1 = x_2 - u$$

$$m\dot{v} = -kx - b(v - v_G) - mg \quad \Rightarrow$$

$$m\frac{d}{dt}(x_2 + v_e) = -k(x_1 + x_e) - b[(x_2 + v_e) - (u + v_{Ge})] - mg$$

$$m\dot{x}_2 = -kx_1 - bx_2 + bu - kx_e - mg$$

Como $kx_e = -mg$

$$m\dot{x}_2 = -kx_1 - bx_2 + bu$$

Resultando no seguinte sistema de equações de estado:

$$\begin{cases} \dot{x}_1 = x_2 - u \\ \dot{x}_2 = \dfrac{1}{m}[-kx_1 - bx_2 + bu] \end{cases}$$

Podemos acrescentar a equação da saída. Como a saída é a deflexão da mola, a partir do equilíbrio, temos que:

$$y = x - x_e \quad \Rightarrow \quad y = x_1$$

Rearranjando as equações:

$$\begin{cases} \dot{x}_1 = +0x_1 + 1x_2 - 1u \\ \dot{x}_2 = -\dfrac{k}{m}x_1 - \dfrac{b}{m}x_2 + \dfrac{b}{m}u \end{cases} \Rightarrow \underbrace{\begin{bmatrix} \dot{x}_1 \\ \dot{x}_2 \end{bmatrix}}_{\dot{\mathbf{x}}} = \underbrace{\begin{bmatrix} 0 & 1 \\ -\dfrac{k}{m} & -\dfrac{b}{m} \end{bmatrix}}_{\mathbf{A}} \underbrace{\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}}_{\mathbf{x}} + \underbrace{\begin{bmatrix} -1 \\ \dfrac{b}{m} \end{bmatrix}}_{\mathbf{B}} \underbrace{u}_{\mathbf{u}}$$

$$y = +1x_1 + 0x_2 + 0u \qquad \underbrace{y}_{\mathbf{y}} = \underbrace{\begin{bmatrix} 1 & 0 \end{bmatrix}}_{\mathbf{C}} \underbrace{\begin{bmatrix} x_1 \\ x_2 \end{bmatrix}}_{\mathbf{x}} + \underbrace{\begin{bmatrix} 0 \end{bmatrix}}_{\mathbf{D}} \underbrace{u}_{\mathbf{u}}$$

O modelo do sistema é da forma:

$$\dot{\mathbf{x}} = \mathbf{Ax} + \mathbf{Bu}$$
$$\mathbf{y} = \mathbf{Cx} + \mathbf{Du}$$

Função de transferência:
Equações do sistema:

$$\begin{cases} \dot{x}_1 = x_2 - u \\ \dot{x}_2 = -\dfrac{k}{m}x_1 - \dfrac{b}{m}x_2 + \dfrac{b}{m}u \end{cases}$$

Aplicando transformada de Laplace:

$$\begin{cases} sX_1 - x_1(0) = X_2 - U \\ sX_2 - x_2(0) = -\dfrac{k}{m}X_1 - \dfrac{b}{m}X_2 + \dfrac{b}{m}U \end{cases}$$

Impondo condições iniciais nulas: $x_1(0) = 0$ e $x_2(0) = 0$

$$\begin{cases} sX_1 = X_2 - U \\ sX_2 = -\dfrac{k}{m}X_1 - \dfrac{b}{m}X_2 + \dfrac{b}{m}U \end{cases}$$

O sistema de equações diferenciais foi transformado em um sistema de equações algébricas. Podemos resolver o sistema de equações algébricas para calcular X_1 em função de U, obtendo a seguinte resposta:

$$X_1 = \frac{-ms}{ms^2 + bs + k}U$$

Como $y = x_1$:

$$Y = \underbrace{\frac{-ms}{ms^2 + bs + k}}_{G(s)}U \Rightarrow Y = G(s)U$$

$$G(s) = \frac{Y}{U} \Rightarrow G(s) = \frac{-ms}{ms^2 + bs + k}$$

A função de transferência $G(s)$ é a relação entre a transformada de Laplace da saída y e a transformada de Laplace da entrada u, considerando condições iniciais nulas.

Análise Transitória

Na análise transitória estamos interessados em observar o comportamento do sistema ao longo do tempo, principalmente a parcela transitória (aquela que diminui e desaparece ao longo do tempo). Um modo de obter a resposta transitória a partir do modelo é a simulação numérica do modelo matemático, ou seja, fazer a integração numérica das equações diferenciais que representam o comportamento do sistema. Isto pode ser feito no Scilab usando-se o comando `csim`, como visto na lista D, ou com o comando ode (lista C), ou usando seu próprio algoritmo de integração (Lista B).

Análise Transitória no Scilab

Exemplo:

```
// Definindo os parametros do sistema:
m=1;b=10;k=900;

// Definindo os polinomios da funcao de transferencia:
// Numerador:
n=(-m)*poly(0,'s','roots');
// Denominador
d=poly([k b m],'s','coef'); //observe a ordem contraria dos coeficientes

// Montando a funcao de transferencia, onde o parametro 'c' indica sistema de
// tempo continuo. Se for um sistema de tempo discreto, use o parametro 'd'.
G=syslin('c',n/d)

// Simulando o sistema para uma entrada degrau (u=0 para t<0 e u=1 para t>0):
// Definindo o vetor tempo:
t=0:0.01:2;

// Definindo a entrada:
u=ones(t);

// Definindo o vetor de condicoes iniciais:
// O sistema é de segunda ordem, logo sao duas condicoes iniciais.
// Não definindo as condicoes iniciais o programa assume como sendo nulas.
x0=[0;0]; // x(0) = 0 e a derivada de x(t) no instante inicial tambem eh nula.

// Realizando a simulacao com o comando csim:
[y]=csim(u,t,G,x0);

// Abrindo uma nova janela de graficos:
xset('window',1)

// Mostrando o resultado da simulacao:
xset('thickness',2)
xset('font size',4)
plot2d(t,y,2)
xtitle('Resposta a degrau','tempo t','Deformacao da mola')
```

****

Observação:

Pode-se definir a função de transferêencia de outra forma:

```
// Montando o polinomio s:
s=poly(0,'s');

// Escrevendo a funcao de transferencia na forma de uma expressao literal f
// onde m, b e k sao os parametros definidos anteriormente e s eh uma
// variavel:
f='[-m*s/(m*s^2+b*s+k)]';

// Calculando a expressao literal substituindo a letra s da expressao f pelo
// seu valor, que eh o polinomio s. Observe que poderiamos ter definido s=3,
// (ou qualquer outro valor) e, neste caso, g não seria um polinomio, mas o
// valor numerico resultante de se substituir 3 no lugar do s da expressao f.
g=evstr(f);

// Montando a funcao de transferencia, onde o parametro 'c' indica sistema de
// tempo continuo. Se for um sistema de tempo discreto, use o parametro 'd'.
G=syslin('c',g)
```

****

A simulação do sistema pode ser feita também usando a representação do sistema no espaço de estados, ou seja, usando as matrizes **A**, **B**, **C**, e **D**:

```
// Definindo os parametros do sistema:
m=1;b=10;k=900;

// Matrizes do sistema:
A=[0 1; -k/m -b/m];
B=[-1;b/m];
C=[1 0];
D=[0];

// Montando o sistema:
suspensao=syslin('c',A,B,C,D);

// Definindo o vetor tempo:
t=0:0.01:2;

// Definindo a entrada:
u=ones(t);

// No espaco de estados temos 2 variaveis de estado:
x0e=[0;0];   // neste caso, x1(0)=0   e   x2(0)=0

// Alem de calcular a saida y, a função csim também permite obter o estado x:
[y,x]=csim(u,t,suspensao,x0e);

// Abrindo uma nova janela de graficos:
xset('window',1)

// Mostrando o resultado da simulacao:
plot2d(t,y,2)
xtitle('Resposta a degrau','tempo t','Deformacao da mola')

// Podemos plotar o grafico do estado x2, por exemplo:
// Abrindo uma nova janela de graficos:
xset('window',2)

// Mostrando o resultado da simulacao:
plot2d(t,x(2,:),2)
xtitle('Resposta a degrau','tempo t','Velocidade da massa')
```

Observação: no caso da simulação usando a função de transferência, as condições iniciais se referem à saída e às suas derivadas. No caso da simulação usando o espaço de estados, as condições iniciais se referem aos estados.

Exercício:
Obtenha as equações de estado e a função de transferência do seguinte sistema, e simule para uma entrada $F(t)$ do tipo degrau (experimente outros tipos de entrada também), considerando a deformação $x(t)$ da mola como saída:



Simule o sistema para diferentes valores de m, c e k, de tal forma que se tenha uma simulação para cada um dos três casos a seguir: $\zeta = \frac{b}{2\sqrt{km}} < 1$, $\zeta = \frac{b}{2\sqrt{km}} = 1$, $\zeta = \frac{b}{2\sqrt{km}} > 1$

Lição de casa:

1 – Considerando o exercício anterior, calcule os autovalores da matriz A e calcule as raízes do polinômio no denominador da função de transferência e compare. Estas raízes (e os autovalores) são os pólos do sistema. Para o caso $\zeta = \dfrac{b}{2\sqrt{km}} < 1$, observe que as raízes (e também os autovalores) são números complexos. Verifique que o módulo deste número complexo é igual à freqüência natural do sistema massa-mola-amortecedor. Verifique ainda que dividindo o módulo da parte real do número complexo pelo módulo do número complexo se obtém o coeficiente de amortecimento. Observe que a freqüência de oscilação é igual ao módulo da parte imaginária do pólo.

2 – Simule o sistema do exercício para entrada nula e diferentes condições iniciais não nulas. Mostre o gráfico de *v* por *x*, e experimente mudar os parâmetros do sistema, tal que se obtenha 3 situações diferentes: pólos complexos, pólos reais e iguais, e pólos reais e distintos. O resultado pretendido são três figuras. Na primeira figura mostre simultaneamente os resultados de diversas simulações com diferentes condições iniciais, mas com os mesmos parâmetros, tais que os pólos sejam complexos. Na segunda figura mostre simultaneamente os resultados de diversas simulações com diferentes condições iniciais, mas com os mesmos parâmetros, tais que os pólos sejam reais e iguais. Na terceira figura mostre simultaneamente os resultados de diversas simulações com diferentes condições iniciais, mas com os mesmos parâmetros, tais que os pólos sejam reais e distintos. Para cada figura construa outra figura mostrando os pólos correspondentes no plano complexo. Observe a ligação entre o comportamento transitório e a posição dos pólos no plano complexo.

3 – Estude a Lista F.